



Graphs in SAS® 9.1.3 and Earlier

Nice. . . but not HDTV!

Graphs in SAS® 9.1.3 and Earlier

- Some graphics options are available.
- Graphs are highly customizable, but not without a lot of code.
- Complex graphs can be created, but some programmers resorted to non-SAS software.

Example 1: Creating Paneled Scatter Plots in SAS® 9.1.3

Extensive coding is required to create paneled scatter plots.

```
proc loess data=sasuser.cars;
ods output OutputStatistics=horse_city;
model horsepower=
run;
proc loess data=
ods output Output
model horsepower=
run;
proc loess data=
ods output Output
model enginesize=
run;
proc loess data=
ods output Output
model enginesize=
run;
quit;
```

```
goptions reset=all nodi
         ypixels=180 ft
pattern1 color=cxffffff
pattern2 color=cxb9cfe7
symbol1 color=cxb9cfe7
symbol2 color=cxb9cfe7
symbol3 color=black val
symbol4 color=cx003178
        value=none widt
proc sort data=horse_c
proc sort data=horse_hi
proc sort data=disp_cit
proc sort data=disp_hig
```

```
axis2 label=(angle=90 rotate=0 "Horsepower");
proc gplot data=horse_city;
 format LowerCL 4.0;
 plot (LowerCL UpperCL DepVar Pred)*mpg_city/ overlay
      vaxis=axis2 name='hp_city';
 run;
quit;
proc gplot data=hor
 format LowerCL 4.
 plot (LowerCL Upp
      overlay vax
 run;
quit;
axis2 label=(angle=
proc gplot data=dis
 format LowerCL 4.
 plot (LowerCL Upp
      overlay vax
 run;
quit;
proc gplot data=disp_highway;
 format LowerCL 4.0;
 plot (LowerCL UpperCL DepVar Pred)*mpg_highway/ overlay
      vaxis=axis2 name='disp_hwy';
 run;
quit;
```

```
filename dan "panel.gif";
goptions display xpixels=480 ypixels=360 gsfname=dan
         gsfmode=replace;
proc greplay nofs tc=sashelp.templt template=l2r2;
igout gseg;
treplay 1:hp_city 2:disp_cty 3:hp_hwy 4:disp_hwy;
run;
quit;
```

(continued)

Example 1: Creating Paneled Scatter Plots in SAS® 9.1.3

The scatter plots that are generated by the previous example code:



Example 2: Creating a Survival Curve in SAS® 9.1.3

The LIFETEST procedure is used to generate a survival curve, as shown in this example code:

```
ods select 'Survival';
proc lifetest data=melanoma plots=(s);
     strata agegrp;
     time time*fail(0);
     run;
quit;
```

(continued)

Example 2: Creating a Survival Curve in SAS® 9.1.3

The basic plot that is generated by the previous example code:



Graphs in SAS® 9.2



ODS Graphics Functionality in SAS® 9.2

- provides statisticians with direct access to statistics in a graphical format
- now enhances over 60 procedures
- requires that you license SAS/GRAPH® software

SAS® 9.2 ODS Graphics: Supported Procedures

Base® SAS Software	SAS/STAT® Software		SAS/QC® Software	SAS/ETS® Software
CORR	ANOVA	MI	ANOM	ARIMA
FREQ	BOXPLOT	MIXED	CAPABILITY	AUTOREG
UNIVARIATE	CALIS	MULTTEST	CUSUM	ENTROPY
	CLUSTER	NPAR1WAY	MACONTROL	ESM
	CORRESP	PHREG	PARETO	EXPAND
	FACTOR	PLS	RELIABILITY	MODEL
	FREQ	PRINCOMP	SHEWHART	PANEL
	GAM	PRINQUAL		SIMILARITY
	GENMOD	PROBIT		SYSLIN
	GLIMMIX	QUANTREG		TIMESERIES
	GLM	REG		UCM
	GLMSELECT	ROBUSTREG		VARMAX
	KDE	RSREG		X12
	KRIGE2D	SEQDESIGN		
	LIFEREG	SEQTEST	SAS® High-Performance Forecasting	SAS® Risk Dimensions®
	LIFETEST	SIM2D		RISK
	LOESS	TCALIS		
	LOGISTIC	TRANSREG	HPF	
	MCMC	TTEST	HPFENGINE	
	MDS	VARIOGRAM		

Example 3: Creating an Enhanced Survival Curve in SAS® 9.2

The following example code uses the ODS GRAPHICS statement to enhance the survival curve:

```
ods html style=normal;
ods graphics on;
ods select 'Product-Limit Survival';
proc lifetest data=melanoma plots=(s);
   strata agegrp;
   time time*fail(0);
run;
quit;
ods html close;
```

Note: The PLOTS=(S) option is not required.

(continued)

Example 3: Creating an Enhanced Survival Curve in SAS® 9.2

The enhanced survival curve that is created by the preceding code:



SAS® 9.2 ODS Graphics: Transforming Plot Attributes

Presto! SAS 9.2 ODS Graphics enables you to easily transform plot attributes.

- ❑ Title
- ❑ Legend
- ❑ Axis label



Using SAS® 9.2 ODS Graphics with the TEMPLATE Procedure

ODS TRACE ON;

PROC <***any-procedure-with-table-or-graph-template***>;


```
Name:      SurvivalPlot
Label:     Survival Curves
Template:  Stat.Lifetest.Graphics.ProductLimitSurvival
Path:      Lifetest.SurvivalPlot
```

(continued)

Using SAS® 9.2 ODS Graphics with the TEMPLATE Procedure

Method 1: PROC TEMPLATE

```
proc template;
   source Stat.Lifetest.Graphics.ProductLimitSurvival;
run;
```

(continued)

Using SAS® 9.2 ODS Graphics with the TEMPLATE Procedure

Method 2:

The SAS Template Window



Example 4: Using SAS® 9.2 ODS Graphics and the TEMPLATE Procedure to Modify the Entry Title in SAS® 9.2

The following example code is used to modify an entry title:

```
proc template;
   define statgraph Stat.Lifetest.Graphics.ProductLimitSurvival;
   dynamic NStrata xName plotAtRisk plotCensored plotCL plotHW
           plotEP labelCL labelHW labelEP maxTime StratumID
           classAtRisk plotBand plotTest GroupName yMin
           Transparency SecondTitle TestName pValue;
   BeginGraph;
      ... more Graph Template Language (GTL)...
      else
         entrytitle "Product-Limit Survival Estimates Over Time";
         ... more Graph Template Language (GTL)...
      EndGraph;

   end;

run;
```

(continued)

Example 4: Using ODS Graphics and the TEMPLATE Procedure to Modify the Entry Title in SAS® 9.2

The survival curve with a modified entry title:



Example 5: Using SAS® 9.2 ODS Graphics and the TEMPLATE Procedure to Modify the Legend and an Axis Label

The following TEMPLATE procedure modifies the text and location of the plot's legend:

```
proc template;
   define statgraph Stat.Lifetest.Graphics.ProductLimitSurvival;
   dynamic NStrata xName plotAtRisk plotCensored plotCL plotHW plotEP
           labelCL labelHW labelEP maxTime StratumID classAtRisk
           plotBand plotTest GroupName yMin Transparency SecondTitle
           TestName pValue;
   BeginGraph;
      ... more Graph Template Language (GTL)...

        /* Default statement:                                    */
        /* DiscreteLegend "Survival" /                           */
        /*                title=GROUPNAME location=outside;      */
      DiscreteLegend "Survival" / location=inside autoalign=(bottom);
      ... more Graph Template Language (GTL)...
   EndGraph;
   end;                                                    (continued)
run;
```

Example 5: Using SAS® 9.2 ODS Graphics and the TEMPLATE Procedure to Modify the Legend and an Axis Label

The second TEMPLATE procedure in this example modifies the label for the Y axis:

```
proc template;
   ...more Graph Template Language (GTL)...
   layout overlay / xaxisopts=(shortlabel=XNAME
                    offsetmin=.05
                    linearopts=(viewmax=MAXTIME))
                      yaxisopts=(label="Survival
                                        Probability in 2000"
                    shortlabel="Survival"
                    linearopts=(viewmin=0 viewmax=1
                    tickvaluelist=(0 .2 .4 .6 .8 1.0)));
   ...more Graph Template Language (GTL)...
run;
```

(continued)

Example 5: Using SAS® 9.2 ODS Graphics and the TEMPLATE Procedure to Modify the Legend and an Axis Label

The plot with the modified legend and axis:



The SAS® 9.2 ODS Graphics Editor and the SAS/GRAPH® Samples Output Gallery

- ODS Graphics Editor—a point-and-click interface that enables you to modify graphs that you create with ODS Graphics. To invoke the editor, use the following ODS statement:

```
ods listing sge=on;
```

- SAS/GRAPH Samples Output Gallery—a gallery of predefined, common graphs.

New SAS/GRAPH® Procedures in SAS® 9.2

- The new functionality dresses up existing procedures and introduces four new procedures:
  - SGPLOT—produces single-cell graphs.
  - SGSCATTER—produces three types of single-cell or multiple-cell scatter plots:
    - gridded
    - comparative
    - matrix
  - SGPANEL—produces multiple-cell classification plots with one or more classification variables.
  - SGRENDER—produces plots from input SAS data sets and ODS graph templates.

Example 6: Creating Single-Cell Graphs with the SGPLOT Procedure in SAS® 9.2

This example code creates a histogram overlaid with two kernel-density plots:

```
ods html file" "sgplot.html"
        style=normal;

proc sgplot data=sashelp.heart;
   histogram cholesterol;
   density cholesterol;
   density cholesterol /
           type=kernel;
run;

ods html close;
```



Example 7: Changing the Overlay and Line Attributes with the SGPLOT Procedure in SAS® 9.2

This example code overlays the two density plots with the histogram:

```
ods html file="sgplot1.html";

proc sgplot data=sashelp.heart;
   density cholesterol;
   density cholesterol /
           type=kernel
           lineattrs=(pattern=dash);
   histogram cholesterol;
run;

ods html close;
```



Example 8: Displaying Outliers with the SGPLOT Procedure in SAS® 9.2

PROC SGPLOT makes showing outliers much easier than is possible with the GPLOT procedure.

```
data cars;
   set sashelp.cars;
   if (msrp >= 100000) then
       expensive=model;
run;

ods html file="sgplot2.html"
         style=normal;

proc sgplot data=cars;
  scatter x=mpg_city y=msrp /
           group=type
           datalabel=expensive;
  loess x=mpg_city y=msrp /
         nomarkers;
run;

ods html close;
```



Example 9: Creating a Grouped Scatter Plot with the SGSCATTER Procedure in SAS® 9.2

In this example, PROC SGSCATTER is used to create a scatter plot with data points and a regression line.

```
ods html file="sgscatter.html";

Title "Auto Performance Statistics";

proc sgscatter data=sashelp.cars;
   where type='Sedan';
   plot (mpg_city weight) *
        (enginesize horsepower) /
        group=origin
   reg=(nogroup clm degree=2) grid
       legend=(noborder);
run;

ods html close;
```



Example 10: Controlling Panel Attributes with the PANELBY Statement in the SGPANEL Procedure in SAS® 9.2

In this example, PROC SGPANEL is used to create default column paneling.

```
title 'Cholesterol by Sex and Age Group';

proc sgpanel data=sashelp.heart;
   format ageatstart age.;
   panelby sex ;
   hbox cholesterol /category=ageatstart;
run;
```



Example 11: Controlling Panel Attributes with the PANELBY Statement in the SGPANEL Procedure in SAS® 9.2

In this example, PROC SGPANEL uses the same data to create panels in rows.

```
proc sgpanel data=sashelp.heart;
   format ageatstart age.;
   panelby sex / columns=1
         layout=rowlattice
         novarname;
   hbox cholesterol /
        category=ageatstart;
run;
```



Example 12: Extracting Graph Template Language from PROCs SGPLOT and SGSCATTER

The TMPLOUT option writes a SAS file of the GTL definition for the graph you have defined in the procedures SGPLOT and SGSCATTER.

```
proc sgplot data=sashelp.class
 tmplout="GTL.sas";
 vbox height / category=age;
run;
```

```
proc template;
 define statgraph sgplot;
  dynamic _ticklist_;
  begingraph;
   layout overlay /
xaxisopts=(type=Discrete
discreteOpts=(tickValueList=
'  _ticklist_));
BoxPlot X=Age Y=Height /
SortOrder=Internal primary=true
LegendLabel="Height" NAME="VBOX";
   endlayout;
  endgraph;
 end;
run;
```

Common ODS Graphics Options

- RESET<=*options*>
- BORDER=ON | OFF
- IMAGEMAP=ON | OFF
- SCALE=ON | OFF
- WIDTH=*dimension*
- HEIGHT=d*imension*
- IMAGEFMT=STATIC | PNG | PS | TIFF| GIF | JPEG (other values are available for this option). See SAS Note 34790, "ODS Statistical Graphics always produces graphics output in the form of a bitmap image."

Generating ODS Graphics Output in Other Formats

ODS Graphics output is available in these formats:

- HTML
- XLS (MSOFFICE2K)
- PDF
- RTF
- PS (Postscript)

Example 13: Using SAS® 9.2 ODS Graphics and the LIFETEST Procedure to Generate Tables

In this example, PROC LIFETEST is used to place the Homogeneity Tests table in RTF format.

```
ods rtf file="lifetest.rtf" style=sasweb;
ods noproctitle;
ods select 'Homogeneity Tests';

proc lifetest data=melanoma;
   strata agegrp;
   time time*fail(0);
run;
quit;

ods rtf close;
```

Stratified Test of Equality over Group			
Test	Chi-Square	DF	Pr > Chi-Square
Log-Rank	0.6882	1	0.4068
Wilcoxon	0.1786	1	0.6726

Example 14: Using the TEMPLATE Procedure to Modify the Pvalue in a Table in SAS® 9.2

In this example, PROC TEMPLATE is used to modify the color and format of the Pvalue in the output table.

```
proc template;
   edit common.ProbChiSq;
   cellstyle _val_ > .35 as {color=red};
   format=6.3;
 end;
run;
```

Test of Equality over Strata			
Test	Chi-Square	DF	Pr > Chi-Square
Log-Rank	1.9623	2	0.375
Wilcoxon	2.3011	2	0.316
-2Log(LR)	2.1493	2	0.341

Example 15: Combining Tables and Graphs in SAS® 9.2 in the RTF Destination

This example illustrates modified, combined output:

```
ods graphics on / reset;
ods rtf file="trafficlighting.rtf" style=sasweb ;
ods select 'Survival Curves' 'Homogeneity Tests';

proc lifetest data=melanoma;
   strata agegrp;
   time time*fail(0);
run;
quit;

ods graphics off;
ods _all_ close;
ods listing;
```

(continued)

Example 15: Combining Tables and Graphs in SAS® 9.2 in the RTF Destination

RTF output that is generated by the preceding code:



Example 15: Combining Tables and Graphs in SAS® 9.2 in the HTML Destination

In addition to using a different destination, this example code changes the Pvalue to a URL.

```
proc template;
  edit common.pvalue;
  cellstyle _val_ as {url="http://support.sas.com"};
  format=6.3;
  end;
run;

ods noptitle;
ods html file="ttests.html" style=sasweb;
ods select 'T-Tests' 'Summary Panel';
ods graphics on / reset;
   proc ttest h0=80 alpha=0.1;
      var time;
   run;
ods html close;
```

(continued)

Example 16: Combining Tables and Graphs in SAS® 9.2 in the HTML Destination

HTML output that is generated by the preceding code:



Code Management

Don't forget to clean up!

```
   /* This TEMPLATE procedure follows the rest of your  */
   /* code.                                             */
proc template;
   delete Common.Pvalue;
   delete Common.Probchisq;
   delete Stat.Lifetest.Graphics.ProductLimitSurvival;
run;
```

Conclusion

SAS 9.2 ODS Graphics capabilities provide new options and functionality that enhance the following tasks:

- ❑ exploring data
- ❑ verifying data
- ❑ publishing data

Resources

SAS Institute Inc. 2009. SAS Note 34790, "ODS Statistical Graphics always produces graphics output in the form of a bitmap image." Cary, NC: SAS Institute Inc. Available at support.sas.com/kb/34/790.html.

SAS Institute Inc. 2009. "SAS/GRAPH Samples Output Gallery." Cary, NC: SAS Institute Inc. Available at support.sas.com/sassamples/graphgallery/index.html.

SAS Institute Inc. 2009. "Secrets of the SG Procedures." Cary, NC: SAS Institute Inc. Available at support.sas.com/resources/papers/proceedings09/324-2009.pdf.

SAS Institute Inc. 2008. "Effective Graphics Made Simple Using SAS/GRAPH® SG Procedures." Cary, NC: SAS Institute Inc. Available at www2.sas.com/proceedings/forum2008/255-2008.pdf.

SAS Institute Inc. 2004. "An Introduction to ODS for Statistical Graphics in SAS® 9.1." Cary, NC: SAS Institute Inc. Available at www2.sas.com/proceedings/sugi29/204-29.pdf.

Note: Additional SAS/GRAPH software technical papers are available at support.sas.com/resources/papers/tnote/tnote_graph.html.